A COMPUTER PROGRAM TO CALCULATE FIRST ORDER EXTINCTION COEFFICIENTS

MURRAY K. KAITTING
Tardis Observatory
533 GlenForrest Blvd.
Waterloo, Ontario N2L 4H9
Canada

Abstract

A description is given of a computer program written in Applesoft BASIC on an Apple II Plus. It is designed to calculate the first order extinction K' by least-squares and to display the observations as functions of air mass.

* * * * *

1. Introduction

Even on a clear night the light of the stars is diminished significantly through absorption and scattering by the earth's atmosphere. The majority of this light loss depends primarily on three parameters, namely: 1) the height of the star above the horizon; 2) the bandpass of the observations; and 3) the atmospheric conditions at the time of the observations.

To model the effects of extinction and the photometric system is very complex indeed. If we calculate the principal extinction (first order) we will have eliminated the largest contributor. The need and the method to determine the principal extinction coefficients will depend primarily on the type of observing program. In all-sky photometry, a set of standard stars scattered throughout the sky is observed to determine K', the extinction coefficient. For inferior sky conditions, a simpler method is appropriate. One needs only to observe a nearby comparison star throughout the night over a range of air masses. The computer program described here uses a different method.

It is generally considered permissable to simply guess a nightly value of K' when doing differential photometry because the comparison star is usually within a degree of the program star under study, and therefore the difference in the air mass of the observation is quite insignificant. This is only true, however, if two conditions are met. First, the two stars must be within one degree of each other, which is not likely for bright stars and a small telescope. Second, the observations of both stars must be made simultaneously, which is difficult. Even if the first criterion is met, the earth's rotation will soon cause the topocentric (apparent) positions of the stars at the individual times of observation to be separated by more than a degree.

These are some of the reasons why it would be better to actually measure the extinction coefficient, K'. It was felt that a linear least-squares determination was not enough to make an intelligent decision on the actual sky conditions. If one simply calculates the K' without plotting the observations as instrumental magnitudes versus air mass, a value for K' will be derived that will be just as dependent on any "poor" observations (i.e., haze moving in) as it will be on those of higher quality. In other words, the value alone will not tell you whether or not the sky remained constant throughout your observations. However, a quick glance at a plot of the observations will.

Values for the graph listed in Table I were taken from observations of a comparison star listed by Henden and Kaitchuck (1982). The author computed the values for local sidereal time. As the observations were reduced to net counts on the star and as this program requires sky count values to run, it was necessary to add one count to each of the observations and input all sky counts as one per second.

Table II shows the calculated values of K' and air mass range. The small discrepancies are probably due entirely to the author's approximation of the local sidereal time of each of the observations. Figure 1 is a plot of these observations.

Table III is a flowchart of the program K Prime, and Table IV contains a listing of the program, along with notes on its customization.

REFERENCE

Henden, A. A. and Kaitchuck, R. H. 1982, *Astronomical Photometry*, Van Nostrand Reinhold Company, Inc., New York.

TABLE I

COMPARISON STAR DATA

Star: α = 02h 07m δ = 40° 23'
Observatory: 39° 33' north latitude

L.S.T.	HA	X	COUNTS PER SEC. V FILTER	SKY
23:35	2:42E	1.165	5661	1
0:24	1:53	1.076	5855	1
1:03	1:14	1.032	5879	1
1:47	0:30	1.005	5888	1
2:16	0:09W	1.001	5898	1
2:21	0:14	1.001	5884	1
3:07	1:00	1.021	5839	1
3:50	1:43	1.062	5656	1
4:31	2:24	1.127	5569	1
5:22	3:15	1.252	5416	1
6:08	4:01	1.423	5188	1
6:38	4:31	1.579	5043	1

TABLE II

K' and Air Mass Range Values

	Henden & Kaitchuk	K Prime Plot
K'v	0.306	0.3
Air Mass Range	0.578	0.571

TABLE III

Flowchart of Program K Prime



TABLE IV

Listing of Program K Prime

COMMENTS ON THE PROGRAM

Although written in Applesoft BASIC, only slight modifications should be necessary for most other computers. The remark statements and accompanying flow charts should facilitate understanding the program.

In order for the program to apply to your observatory location, change Line #360 to the latitude of your observatory.

If you are using pulse counting electronics, change Line #370 to your determined value for deadtime correction.

For those doing DC photometry, the program is not difficult to alter. Simply delete the following Lines: #370, #460, #660, and #720, and change Line #740 to Cl = (A-B).

```
PROGRAM: FIRST ORDER EXTINCTION


110  LOMEM: 16384
120  ONERR  GOTO 2220
130  TEXT : CLEAR : HOME
140  DIM X(18),Q(18)
150  VTAB (5)
160  HTAB (7): INVERSE : PRINT "FIRST ORDER EXTINCTI
ON PLOT": NORMAL
170  PRINT
180  PRINT  TAB( 19)"BY": PRINT
190  PRINT  TAB( 11)"MURRAY K. KAITTING"
200  PRINT
210  HTAB (11): INVERSE : PRINT "TARDIS OBSERVATORY"
: NORMAL
220  PRINT
230  PRINT  TAB( 14)"WATERLOO,ONT."
240  PRINT  TAB( 17)"CANADA"
250  PRINT  TAB( 17)"(1983)"
260  PRINT
270  PRINT  TAB( 3)"A SET OF 18 OBSERVATIONS UP TO 3
 AIR"
280  PRINT  TAB( 3)"MASSES IS ALLOWED WITH THIS  PRO
GRAM"
290  PRINT
300  PRINT " ALL OBSERVATIONS ARE IN HR,MIN OR SEC"
310  PRINT "----------------------------------------
"
320  PRINT
330  PRINT
340  REM DEFINE CONSTANTS
350 RD = (4 *  ATN (1)) / 180
360 LA = 43.433 * RD
370 DT = 1.2E - 7
380 A$ = "AIR MASS RANGE= "
390  PRINT  TAB( 6)"INPUT MODE(UBV) ?"
```

```
400  GET M0$
410  IF M0$ <  > "U" AND M0$ <  > "B" AND M0$ <  > "
V" THEN  GOTO 130
420  HTAB (8): INPUT "RA. OF STAR===> ";R1,R2
430  HTAB (7): INPUT "DEC. OF STAR===> ";R3,R4
440 RA = (R1 + (R2 / 60))
450 DC = (R3 + (R4 / 60)) * RD
460  HTAB (5): INPUT "INTEGRATION TIME=> ";IT
470  PRINT
480  PRINT "NUMBER OF INTEGRATIONS"
490  INPUT "     PER OBSERVATIONS=> ";IN
500  PRINT
510  PRINT  TAB( 6)"TOTAL NUMBER OF"
520  HTAB (7): INPUT "OBSERVATIONS===> ";OB
530  FOR G = 1 TO OB
540 A = 0:B = 0:C1 = 0:CA = 0:CB = 0:M1 = 0
550  PRINT
560  HTAB (13): INPUT "L.S.T.===> ";SA,SB
570  REM COMPUTE HOUR ANGLE
580 HA = (((SA + (SB / 60)) - RA) * 15) * RD
590  REM ADD COUNTS & NORMALIZE
600  REM TO COUNTS PER. SECOND
610  FOR I = 1 TO IN
620  HTAB (15): INPUT "STAR===> ";AC(I)
630 A = A + AC(I)
640  NEXT I
650 A = A / IN
660 A = A / IT
670  FOR J = 1 TO IN
680  HTAB (16): INPUT "SKY===> ";BS(J)
690 B = B + BS(J)
700  NEXT J
710 B = B / IN
720 B = B / IT
730  REM CORRECT FOR DEAD-TIME
740 C1 = (A - B) * (1 + (A - B) * (DT))
750  REM CALC RAW INST. MAG
760 Q(G) =  - 2.5 * ( LOG (C1) /  LOG (10))
770  REM CALC AIRMASS OF OBSERVATIONS
780 CA =  SIN (LA) *  SIN (DC)
790 CB =  COS (LA) *  COS (DC)
800 M1 = (CA + (CB *  COS (HA))) ^  - 1
810 X(G) = M1 * (1 - (0.0012 * ((M1 ^ 2) - 1)))
820  NEXT G
830  REM SORT OBS. IN ORDER OF
840  REM INCREASING AIRMASS
850 N = OB:M = N
860 M =  INT (M / 2)
870  FOR J = 1 TO N - M:I = J
880 L = I + M
890  IF X(I) <  = X(L) THEN  GOTO 920
900 T = X(I):X(I) = X(L):X(L) = T:I = I - M
910  IF I > 0 THEN  GOTO 880
920  NEXT J
930  IF M > 1 THEN  GOTO 860
940  FOR I = 1 TO N:X(I) = X(I): NEXT I
950  REM CALC AIRMASS RANGE OF
960  REM       OBSERVATIONS
970 AR = X(N) - X(1)
980 AR =  INT (AR * 1000) / 1000
990 S1 = 0:S2 = 0:S3 = 0:S4 = 0
1000  PRINT "WOULD YOU LIKE A COPY ? (Y/N)"
1010  GET AN$
1020  IF AN$ <  > "Y" AND AN$ <  > "N" THEN  GOTO 10
00
```

```
1030  IF AN$ = "N" THEN  GOTO 1180
1040  INPUT "STAR NAME ";SN$
1050  INPUT "DATE ";DY,MT,YR
1060  PR# 1
1070 S$ = " STAR: "
1080  PRINT  TAB( 27)"TARDIS OBSERVATORY"
1090  PRINT  TAB( 27)"******************"
1100  PRINT S$;SN$
1110 D$ = "DATE: "
1120 RA$ = " RA: ":DC$ = "DEC: "
1130  PRINT  TAB( 3)RA$;R1;"HR ";R2;" MIN "; TAB( 26
)"1ST. ORDER EXTINCTION"; TAB( 15)D$;DY;"/";MT;"/";Y
R
1140  PRINT  TAB( 3)DC$;R3;" DEG ";R4;" MIN "
1150  PRINT "----------------------------------------
-------------------------------"
1160  PR# 0
1170  REM CALC K',LEAST-SQUARES
1180  FOR R = 1 TO OB
1190 S1 = S1 + X(R)
1200 S2 = S2 + X(R) ^ 2
1210 S3 = S3 + Q(R)
1220 S4 = S4 + Q(R) * X(R)
1230  NEXT R
1240 K = ((S1 * S3 - OB * S4) / (S1 ^ 2 - OB * S2))
1250  IF K > 0 THEN  GOTO 1270
1260  IF K <  > 0 THEN K = K *  - 1
1270 K =  INT (K * 100) / 100
1280  HCOLOR= 7
1290  HGR
1300  REM ##############
1310  REM # PLOT GRAPH #
1320  REM ##############
1330  HPLOT 0,0 TO 278,0 TO 278,159 TO 0,159 TO 0,0
1340 H = 10
1350  HPLOT 0,H TO 2,H
1360 H = H + 15
1370  IF H > 159 THEN  GOTO 1390
1380  GOTO 1350
1390  HPLOT 145,159 TO 145,157
1400  REM SCALE X-AXIS
1410  FOR R = 1 TO OB
1420 X(R) = ((X(R) * 140) - 140)
1430  NEXT R
1440  REM SORT MAG. IN ORDER OF
1450  REM   DECREASING MAG.
1460 N = OB:M = N
1470 M = M
1480 M =  INT (M / 2)
1490  FOR J = 1 TO N - M:I = J
1500 L = I + M
1510  IF Q(I) <  = Q(L) THEN  GOTO 1540
1520 T = Q(I):Q(I) = Q(L):Q(L) = T:I = I - M
1530  IF I > 0 THEN  GOTO 1500
1540  NEXT J
1550  IF M > 1 THEN  GOTO 1480
1560  FOR I = 1 TO N:Q(I) = Q(I): NEXT I
1570  FOR R = 1 TO N
1580 Q(R) = Q(R) *  - 1
1590  NEXT R
1600  REM CALC RANGE OF Y-AXIS
1610 MG = Q(1) - Q(N):MG = MG * 100
1620  REM CALC INDIVIDUAL DELTA MAG
1630 D1 = Q(1):D2 = Q(1) - Q(2):D3 = Q(1) - Q(3):D4 =
```

```
Q(1) - Q(4):D5 = Q(1) - Q(5):D6 = Q(1) - Q(6):D7 = Q
(1) - Q(7):D8 = Q(1) - Q(8):D9 = Q(1) - Q(9)
1640 E1 = Q(1) - Q(10):E2 = Q(1) - Q(11):E3 = Q(1) -
Q(12):E4 = Q(1) - Q(13):E5 = Q(1) - Q(14):E6 = Q(1) -
Q(15):E7 = Q(1) - Q(16):E8 = Q(1) - Q(17):E9 = Q(1) -
Q(18)
1650  REM ################
1660  REM # SCALE Y-AXIS #
1670  REM ################
1680 P1 = 3:P2 = D2 * 150:P3 = D3 * 150:P4 = D4 * 15
0:P5 = D5 * 150:P6 = D6 * 150:P7 = D7 * 150:P8 = D8 *
150:P9 = D9 * 150
1690 Z1 = E1 * 150:Z2 = E2 * 150:Z3 = E3 * 150:Z4 =
E4 * 150:Z5 = E5 * 150:Z6 = E6 * 150:Z7 = E7 * 150:Z
8 = E8 * 150:Z9 = E9 * 150
1700 Q(1) = 3:Q(2) = P2:Q(3) = P3:Q(4) = P4:Q(5) = P
5:Q(6) = P6:Q(7) = P7:Q(8) = P8:Q(9) = P9:Q(10) = Z1

1710 Q(11) = Z2:Q(12) = Z3:Q(13) = Z4:Q(14) = Z5:Q(1
5) = Z6:Q(16) = Z7:Q(17) = Z8:Q(18) = Z9
1720  REM PLOT THE OBSERVATIONS
1730  IF X(N) > 279 THEN  GOTO 2250
1740  FOR R = 1 TO OB
1750 Q(R) = Q(R) + 10
1760  NEXT R
1770  FOR R = 1 TO OB
1780 X(R) = X(R) + 5
1790  NEXT R
1800  FOR R = 1 TO OB
1810  HPLOT X(R),Q(R)
1820 Q(1) = 10
1830  IF Q(N) > 158 THEN Q(N) = Q(N) - 3
1840  NEXT R
1850  REM PLOT FIRST OBSERVATION
1860  HPLOT X(1),Q(1)
1870  HPLOT X(1) - 1,Q(1) - 1
1880  HPLOT X(1) + 1,Q(1) + 1
1890  HPLOT X(1) + 1,Q(1) - 1
1900  HPLOT X(1) - 1,Q(1) + 1
1910  REM PLOT OTHER OBSERVATIONS
1920  FOR R = 2 TO OB
1930  HPLOT X(R) - 1,Q(R) - 1
1940  NEXT R
1950  FOR R = 2 TO OB
1960  HPLOT X(R) + 1,Q(R) + 1
1970  NEXT R
1980  FOR R = 2 TO OB
1990  HPLOT X(R) + 1,Q(R) - 1
2000  NEXT R
2010  FOR R = 2 TO OB
2020  HPLOT X(R) - 1,Q(R) + 1
2030  NEXT R
2040  HOME : VTAB (21)
2050  REM FORMAT DATA TO PRINTER
2060  PRINT  SPC( 42)"1"; SPC( 4)"(AIRMASS)"; SPC( 3
)"2"; SPC( 15)"3"
2070 AR = (X(OB) / 140) - (X(1) / 140):AR =  INT (AR
 * 1000) / 1000
2080  PRINT  SPC( 2)"K";: PRINT M0$;: PRINT "=";: PRINT
K;: PRINT  SPC( 7)A$;AR
2090  IF AN$ = "N" THEN  GOTO 2140
2100  PR# 1
2110  PRINT  CHR$ (9);"GME"
2120  PR# 0
2130  REM FORMAT DATA TO MONITOR
```

```
2140  PRINT "1"; SPC( 4)"(AIRMASS)"; SPC( 6)"2"; SPC(
17)"3"
2150  PRINT  TAB( 3)"K";: PRINT M0$;: PRINT "=";: PRINT
K;: PRINT  TAB( 17)A$;AR
2160  PRINT  TAB( 3)"FOR ANOTHER OBSERVATION PRESS (
R)"
2170  GET R$
2180  IF R$ = "R" THEN  GOTO 130
2190  TEXT : HOME
2200  VTAB (10)
2210  HTAB (13): FLASH : PRINT "10-4 GOOD BUDDY": NORMAL

2220  REM
2230  REM DATA IS OUT OF PROGRAM LIMITS
2240  REM
2250  TEXT : HOME : VTAB (6): HTAB (5): FLASH : PRINT
"YOUR VALUES EXCEED 3 AIR-MASSES": NORMAL
2260  PRINT : PRINT
2270  HTAB (13): INVERSE : PRINT "OR ARE UNPHYSICAL!
": NORMAL
2280  PRINT
2290  PRINT  SPC( 3)"REMEMBER,YOUR CONFIDENCE DIMINI
SHES"
2300  PRINT
2310  PRINT  SPC( 6)"AS THE INVERSE-SQUARE OF THE"
2320  PRINT
2330  PRINT  SPC( 15)"AIR-MASS !"
2340  PRINT
2350  HTAB (10): INVERSE : PRINT "LETS GET REAL BUDD
Y !": NORMAL
2360  END
```

TARDIS OBSERVATORY
****

STAR:
RA: 2HR 7 MIN — 1ST. ORDER EXTINCTION — DATE: 23/12/1983
DEC: 40 DEG 23 MIN



Figure 1. The observations taken from Table I plotted as instrumental magnitudes versus air mass. The y-axis is incremented in tenths of a magnitude.